
1.   Background: shell languages

=========================

A shell programming language—also known as a scripting language—is a specialized programming language that is useful for automating administrative procedures.  A good shell language, for example, will allow a user to do tasks like

· conditionally run a series of programs, like you'd conditionally run statements in a regular language;

· feed input to these programs; and

· store output from these programs in files, or send that output to printers.

Shell languages have four typical characteristics:

· Shells are tiny languages.   They provide the user to manipulate variables and files, using a small set of built-in commands.  These commands typically include assignment statements, looping commands, conditional commands (e.g., if-then-else), and possibly procedures.

· Shells are, however, powerful languages.  Any statement that a shell fails to recognize as a built-in command is treated as a command-line-style command, and evaluated accordingly.  A statement, for example, like “copy file.txt file.bak” can be used as a step in a shell code.  Shell languages also support tests that check for the successful completion of a command like “copy”, which support the use of logic to recover from failures.

· Shell languages are typically easy to code in.  This ease of coding stems from the power of shells, as well as the use of interpretation to execute shell code.  Here, “interpreted” means that shell code is executed “as is”, instead of being compiled—translated, in its entirety, into a low-level language before being executed.  Skipping the initial translation makes it much easier to create scripts one line at a time using a text editor.

· Shell languages, however, are typically slower than compiled languages.  This loss of speed is also due to the failure to compile shell code before running it.  This loss of speed, however, is typically not critical in situations where the user is trying to automate a procedure that involves executing a sequence of programs—again, the sort of problem that shells handle well.

Shell languages come in a variety of shapes and forms.  Some of the more common shells include

· sh-  the original Unix shell, also known as the Bourne shell

· csh-  the second Unix shell, created by Berkeley  (pronounced "c shell")

· ksh-  the "Korn" shell

· bash-  the "Bourne again" shell, a shell that attempted to incorporate features from sh, csh, and ksh

· dosshell- the built-in Microsoft scripting language that shipped with MS/DOS, and persists in Windows 2000.

2.   An Introduction to the DOSshell command language

==========================================

The Microsoft shell language, DOSshell, is, at best, an adequate shell language.  Given a choice between DOSshell and a port of bash to the Windows environment, most experienced programmers that I’ve spoken to would use bash, hands-down.  DOSshell, however, is a standard part of the Microsoft environment.    DOSshell, to its credit, is adequate for doing most routine system administration tasks—especially if one is willing to augment DOSshell with simple programs that support some of the less commonly used, but useful, shell language features that DOSshell doesn’t provide. 

The balance of this section covers basic DOSshell constructs in more detail.  Constructs covered include DOSshell comments; shell-driven program execution; the pause command; variables and variable-updating commands; shell script parameters; “if” statements; command status checking; file-system-object status checking; “for” loops; text file manipulation; getting input from the user; scoping; and the “help” command.

2.1  Batch Files

-------------------

DOS Shell scripts are created as text files, with the suffix .BAT.  DOS Shell scripts are invoked by entering the name of the script to execute at a DOS command prompt:  e.g., 

C:\>  FOO

Here, the file FOO.BAT is presumed to be in the current directory, or in some other directory that the DOS command interpreter routinely searches when interpreting the name of executable commands.

2.2  Comments

------------------

Comments in the DOS Shell language are proceeded by the rem command.  Some implementations of DOS Shell also recognize // as a comment character, but commenting with // is nonstandard-- DOS under NT 4.0, for example, complains.  

· Example:  the following lines do nothing:

  rem this is a comment

  rem so is this

2.3    Shell-driven program execution

---------------------------------------------

A statement that simply runs a command is created by simply entering the name of that command, followed by any arguments.   

· Example:  the following lines invoke the date and dir commands:

  date

  dir c:\temp

Commands can also be run in ways that suppress command echo and command output.

If you don't want commands to echo—that is, if you don't want the command lines themselves to print as they execute— one of two constructs to suppress echo.

· Construct 1:  echo off/ echo on

· Example: the following one-line command file will print ">dir c:\temp", followed by the listing:

   dir c:\temp

The  following three-line command file will simply list c:\temp, without echoing the “dir” command:

   echo off

   dir c:\temp

   echo on

The reason for issuing the followup "echo on" is that, without the "echo on", all subsequent echoing of command, including the C: prompt, will simply be suppressed.  Working with a command window that doesn't have a command prompt can get a little unsettling.

· Construct 2:  placing an "@" in front of a line suppresses the echo for just that line

· Example:  the following one-line command file has the same effect as the sample three-line command file that uses echo off/echo on:

   @dir c:\temp

Suppressing command echo is useful when you don't want to confuse your user or clutter the screen by showing that person what all is going on in your command file, from command to command. 

· Example:  To communicate a simple message to the user without confusing the situation with command echo, I recommend using a statement like

   @echo this is the message that the user should see

If you also want to suppress a command's output, redirect that output to the nul file.

· Example:  the following command copies the file foo.txt to c:\ "silently"; copy's informational message about what files were copied will be not displayed on the terminal:

   @copy foo.txt c:\ >nul

2.4   Holding output on the screen

-----------------------------------------

At times, it can be useful to keep output from a command file from flashing by too quickly.  If this is indeed so, use the pause command to achieve your goal

   pause

When you issue the pause command, the system runs a program that does nothing until the user presses a key.  Running a program that does nothing until a key press freezes the output on the screen until the key press is effected.

2.5   Defining and using DOSshell variables
-----------------------------------------------------

A variable is a named quantity that corresponds to some value.  In DOSshell, all variables—even variables that look like numbers—are defined as string variables, or text.

String variables, in DOSshell, are defined and assigned values, using the set command.

· Examples:  the following lines define environment variables named datecommand and workingdir

  set datecommand=date

  set workingdev=c:\

  set workingdir=\temp

To use string variables, simply enclose them in % signs.

· Examples:  the following lines of code invoke the "date" and "dir c:\" commands, assuming the environment variable assignments given above:

  %datecommand%

  dir %workingdev%%workingdir%

It's good practice to define environment-sensitive parameters at the top of a script using set commands, to give others that follow you a clear sense of just what your batch program expects from its operating environment.

One potentially annoying feature of DOSshell is that variables outlive the scripts that create them.  If I, for example, invoke a command prompt window; run a DOSshell in that prompt window that defines a variable named %foo%; and exit the DOSshell, %foo% will hang around after the shell completes execution.

· Example:  If the shell script setfoo.bat consists of one line,  

set foo=hello

then the last command in the following sequence of commands should print “hello”:

C:\>  setfoo

C:\>  echo %foo%

To avoid cluttering the environment with useless definitions, use the setlocal/endlocal commands to restrict the scope of definition in a shell script:

· Example:  after the following five-line command file finishes executing, the environment variable persistent stays defined;  all other environment variables are not.

    set persistent=this stays around

    setlocal

    set transient1=this disappears

    set transient2=this too shall pass

    endlocal

WARNING:  Some implementations of DOSshell impose short (i.e., 80 character) maximum lengths on variables.  The only effective strategy for creating logic that operates on more than the allowed number of characters under these versions of DOS is to work around the variable-length limitations—e.g., by using more than one variable to accomplish a task.

2.6   Using command line parameters

---------------------------------------------

A command file can be started with command-line parameters: auxiliary arguments that can be used, by a coder, to qualify (parameterize) the operation of a shell script:


C:\>  myscript  a   b   c

If you write a script that expects to use command line parameters, you can use nine special variables, %0 ... %9, to access these parameters.   The first variable, %0, gives you the name by which a batch file is invoked.   Subsequent variables, %1 ... %9, give you the command file's first nine parameters.

· Example:  the following command file prints the name with which it was invoked and its first three parameters:

    @echo this command file was invoked as %0

    @echo its first parameter was %1

    @echo its second parameter was %2

    @echo its third parameter was %3

If the above command file is named STUFF.BAT and invoked as  STUFF A B C ,  it will print

    this command file was invoked as STUFF

    its first parameter was A

    its second parameter was B

    its third parameter was C

Missing command line parameters are passed to batch files as null strings.  If, for example, the above command file is named STUFF.BAT and invoked as  STUFF A,  it will print

    this command file was invoked as STUFF

    its first parameter was A

    its second parameter was

    its third parameter was

2.7   Conditional execution of code

-------------------------------------------

The DOSshell "if" statement is used to test whether a condition is true.  "If" is very much like the old unstructured Basic "if”: when "if" evaluates to true, it gives you the option of executing exactly one command, which can (and often is) a goto.

The basic test operators for “if” conditions are == (string equality), and two others discussed below.   The not operator can also be used to change the sense of the test.

· Example:  the following command file runs a rough check on how many parameters it has.   Note the use of the quotes around the parameter variables to handle the case when these variables are null:

      if "%1"=="" @echo this command file was called with no parameters

      if "%1"=="" @goto exit

      if "%2"=="" @echo this command file was called with one parameter

      if "%2"=="" @goto exit

      @echo this command file was called with two or more parameters

rem not  the colon in front of the word “exit” to flag this identifier as a label—a target of a goto

rem

:exit

· Example:  the following logic does exactly the same thing as the logic above, but using NOT:

      if NOT "%3"=="" @echo this command file was called with two or more parameters

      if NOT "%3"=="" @goto exit

      if NOT "%2"=="" @echo this command file was called with one parameter

      if NOT "%2"=="" @goto exit

      @echo this command file was called with no parameters

      :exit

WARNING:  Some implementations DOSshell will **not** allow you to embed a 'noecho' (@) character in front a command that trails an "if".  If the DOS batch file processor that   you are using does not support @ in the middle of a command, you will see an error message that complains about an invalid command at the offending line.

2.8    Checking a command's status

-------------------------------------------

A status code is a value that indicates whether that program succeeded or failed—and possibly how badly that program failed, if it indeed failed.  Programs in most common languages, including DOSshell, can be coded in ways that cause them to generate a status code when they terminate.  In C and C++ codes, for example, the exit() procedure, which causes a program to stop, can be supplied with a value that is turned into a status code.

Status codes generated by programs that execute in command windows are made available to DOSshell immediately on program exit.  These values can be used, by DOSshell “if” commands, to check how cleanly a program ran—and to take corrective action on program failure.  The key construct is "errorlevel".  "if errorlevel n" evaluates to true if the error level of the last command was less than or equal to n-- and false otherwise.

· Example:  the following command file runs a program named foo.exe in the current directory, and then reports on foo's exit status.

        foo

        if errorlevel 0 goto error0

        if errorlevel 1 goto error1

        goto error2

    :error0

        @echo the error status was 0

        goto exit

    :error1

        @echo the error status was 1

        goto exit

    :error2

        @echo the error status was at least 2

        goto exit

    :exit

It is traditional, in DOS, to return 0 if a command succeeds, and to return increasingly larger numbers-- 1, 2, 3, etc.-- as the failure type becomes more severe.

In my batch files, I have used simple, auxiliary command utilities of my own creation to check variables in ways that standard Windows utilities don't provide-- for example,

· whether a string defined by a variable comes before a string defined by second variable, in alphabetical order

· whether a variable names a readable, nonempty file

· whether a variable names a directory

· whether a variable names a file that was created after a certain date

I invoke these auxiliary utilities with command line parameters—e.g., a command that checks if one string is less than a second, using a program named str-leq that I created, might read:

 
str-leq  %string1% %string2%

I then use errorlevel to select the follow-on logic accordingly.

2.9    Checking a file system object's status

----------------------------------------------------

The DOS batch file language has one imperfect test for checking if a file system object "name.ext" exists:

   if exist name.ext

If "if exist" evaluates to true, then name.ext corresponds to a file or a directory.

WARNING:   The DOS documentation claims that you can also tell whether a file system object is a file or a directory by appending "\nul" to the end.  Ostensibly

     if exist name.ext\nul

evaluates to false when name.ext is a file, and true when name.ext is a directory.    This test, however, is somewhat broken, since the logic only works correctly when  the string being tested contains no more than one backslash.  More specifically,

     if exist dir\name.ext\nul

always evaluates to true if dir\name.ext exists, regardless of whether dir\name.ext is a directory.  (Ugh!)   I worked around this problem in one situation by creating my own auxiliary DOS command that checks to see if a name is a directory.  Other checks of a file's attributes would require the creation of other auxiliary utilities that do similar tasks.

WARNING:  In some versions of DOS, the "exist" test apparently fails to start searching for a file in the current directory!  If the DOS batch file interpreter on your system evaluates

       if exist foo.bar

to false when foo.bar is a file system object in the current directory,  use the following form of the name instead:

       if exist .\foo.bar

This initial .\ will force DOSshell to check in the current directory for the object in question.

2.10    Invoking other batch files

----------------------------------------

To transfer control completely to another batch file, use the name of that batch file as an ordinary DOS command.  Control will transfer, and never return.

· Example:  the following batch file transfers control to bar.bat.   Bar.bat is called with two parameters-- the environment variables param1 and param2.   The line below the call to bar.bat—the dir command—will never execute.

      if not exist bar.bat @echo can't find the bar batch file

      if not exist bar.bat goto exit

      bar %param1% %param2%

      REM the following line is dead code

      dir c:\

   :exit

To invoke another batch file as a subroutine, use the "call" command.

· Example:  the following command file transfers control to bar.bat, which then returns.  Bar.bat is called with two parameters-- the environment variables param1 and param2.

      if not exist bar.bat @echo can't find the bar batch file

      if not exist bar.bat goto exit

      call bar %param1% %param2%

      REM the following line **will** execute before completion-- if bar succeeds.

      dir c:\

   :exit

NOTE:  One of DOSshell’s crucial weaknesses is its inability to support compound statements  (i.e., begin/end blocks).  The ability to call a second shell from within a first is sometimes the only way to create a code where one statement to trigger the execution of multiple commands  (cf. FOR loops, below).

2.11   Using more than nine batch file parameters

------------------------------------------------------------

A batch file that is called with more than nine parameters can use the shift command to access the parameters that are "out to the right".  Shift copies %1 to %0, %2 to %1, %3 to %2, etc., with %9 being set to the old tenth parameter.

· Example:  The following loop prints out  all of a batch file's parameters:

      :loop

           if "%1"=="" goto endloop

           @echo %1

           shift

           goto loop

      :endloop

2.12   For loops

--------------------

The final construct to be covered here is the DOS shell for loop, which is written in a manner similar to the following:

    for %%p in ( thing1 thing2 thing3 ) do command

As was the case with "if", "command" is any single DOS command, including a second for loop.  On every iteration of the loop, the loop variable—here, %%p—takes on a new value.

· Example:  The following for loop prints the numbers from 0 to 9

    for %%p in ( 0 1 2 3 4 5 6 7 8 9 ) do echo %%p

· Example:  The following one-line batch file calls bar.bat 100 times with all possible combinations of two digits:

    for %%p in ( 0 1 2 3 4 5 6 7 8 9 ) do for %%q in ( 0 1 2 3 4 5 6 7 8 9 ) do call bar %%p %%q

IMPORTANT:  Note the double percent sign before the p, as well as the use of a single letter as an index variable.  Both conventions are important, and must be followed.

WARNING:   Some versions of DOSshell will not allow you to embed a 'noecho' (@) character in front a command that trails a "for".  If the DOS batch file processor that you are using does not support @ in the middle of a command, you will see an error message  that complains about an invalid command at the offending line.

WARNING:  Some versions of the DOS batch language will not allow you to nest for loops.  To change a batch file that contains a nested for loop into one without a nested loop,  use another batch file and a call to that file.  

· Example:  Assume that your initial file, foo.bat, contains the following line--

       for %%p in ( 0 1 2 3 4 5 6 7 8 9 ) do for %%q in ( 0 1 2 3 4 5 6 7 8 9 ) do call bar %%p %%q

Assume, furthermore, that you either need--or wish--to convert this logic into equivalent logic that does not contain a nested for loop.  To do the conversion,

· first create a batch file--say, called auxil.bat--that contains the following code

       for %%q in ( 0 1 2 3 4 5 6 7 8 9 ) do call bar %1 %%q

· after you create auxil.bat, modify the original code in foo.bat to read

       for %%p in ( 0 1 2 3 4 5 6 7 8 9 ) do @call auxil %%p

2.13    For loops and text file manipulation

----------------------------------------------------

The FOR command’s /F switch also allows the user to read and process  files.   The FOR /F command, by default, reads the file a line at a time, skipping blank lines, and assigns the first “word” on the line to the FOR loop variable.  Here, the term “word” means the initial sequence of characters, up to the first blank.  For example, if the file foo.txt contained the text

a,b c;d

k,l;m n

then the command

FOR /F %%i in ( foo.txt ) DO @echo %%i # %%j # %%k

would generate the output

a,b  #  %j  #  %k

k,l;m  #  %j  #  %k

Two options for the FOR command allow the user to change the number and the positioning of words being captured, and to specify characters other than spaces as word separators:

· The first of these options, tokens, accepts a list of the positions to be captured.  Second and subsequent words captured by the “tokens” option are automatically assigned to the second and subsequent index ( %% ) variables maned on the command line.   

· The second, delims, accepts a list of word-ending characters.  

For example, given the earlier description of the content, for foo.txt, the command

FOR /F "tokens=1,2,3 " %%i IN ( foo.txt ) DO @echo %%i # %%j # %%k

would generate the output

a,b  #  c;d  # 

k,l;  # m,n  # 

while 

FOR /F "tokens=1,2,3 delims=, " %%i IN ( foo.txt ) DO @echo %%i # %%j # %%k

would generate

a   #   b  #  c;d 

k  #  l;m  #  n

Other options for the FOR command support wildcard lookups on file names; recursive directory traversal; access to file characteristics; and generation of sequences of numbers.  See the help documentation on FOR commands for details.

2.14     Getting input from the user

--------------------------------

The following trick for getting input from the console is based on one suggested by a DOS batch shell website, http://members.nbci.com/_XMCM/batfiles/index.htm:

@echo please enter your data, followed by <enter>, then ^Z

@COPY CON foo.txt >nul

FOR /F "tokens=1,2,3 " %%I IN ( foo.txt ) DO @ECHO %%I # %%J # %%K

Note the use of the console pseudo-device, CON, to obtain input, and the use of the copy command to save the input to a file.  The ^Z is needed to signal EOF to the copy command.

2.15   Limiting the scope of set commands to a batch file run

-------------------------------------------------------------------------

Normally, the effects of the "set" command persist beyond the end of a batch file execution.  If you wish to limit the effect of set to within a batch file, use the setlocal and endlocal batch commands.

· Example:  in the following batch file, the definition of the environment variable named persistent persists after the batch file completes;  the definitions of transient1 and transient2 are removed when the batch file finishes execution.

      set persistent=this definition will persist

      setlocal

      set transient1=this definition shall pass

      set transient2=this too shall pass 

      endlocal

2.16  Getting more help

----------------------------------

This primer is meant merely to introduce basic DOS commands—not to exhaustively describe them.  A final DOS command, help, is a good starting point for learning more about what other DOS commands are available.  So are various web-based resources on DOS shell.  Searching for words like  “DOS”, “BATCH”, “SETLOCAL” and “PRIMER” in any major search engine is a good way to locate Web-based online help.

3.   An Incomplete Style Guide for Creating DOS Shell Scripts

==============================================

Some favorite guidelines for coding shell scripts in general (DOS scripts, in particular):

· Tell the user how the script is supposed to work in the script header.   Specify 

· the script's purpose;  

· the meaning and form of any inputs;  

· the meaning and form of any outputs; and 

· the specific effects of any actions that the script performs. 

· Be paranoid about all circumstances beyond your control

· If your script expects to see command line parameters, use explicit checks to ensure that those parameters are present:  e.g., 

     
@IF "%1"=="" @Echo No First Parameter!

@IF "%1"=="" @Goto exit

· Similarly, if your script expects to read input files, use explicit checks to ensure that those files are present:  e.g.,

@IF NOT EXIST %1 @Echo Specified List of Students ( %1 ) Doesn't Exist!

@IF NOT EXIST %1 @Goto exit

· At times, checking the status of every command being executed might be the right thing to do.  At the very least,

· if action B requires the successful completion of action A, then

· verify that A completed successfully before moving on to B.

· Take pains to create meaningful, self-describing error messages:  i.e., use

      

@IF NOT EXIST %1 @Echo Specified List of Students ( %1 ) Doesn't Exist!

 instead of

 @IF NOT EXIST %1 @Echo File Doesn't Exist!

· Hide unimportant output

· Send the output from commands whose output doesn't need to be displayed to the "bit bucket".  Internal DOS commands and commands that manipulate intermediate files are good examples of commands whose output the user probably doesn’t need to know.  Examples:

· Use

 
     @IF NOT EXIST %1 @Echo Specified List of Students ( %1 ) Doesn't Exist!

instead of 

      IF NOT EXIST %1 Echo Specified List of Students ( %1 ) Doesn't Exist!

or even

      IF NOT EXIST %1 @Echo Specified List of Students ( %1 ) Doesn't Exist!

· Always use

       @REM 

 rather than 

       REM

if echo is turned on (which it should usually be, anyway).  No one should have to be told that they're staring at comments.  If you want to show a message to the operator, use a comand like

       @ECHO this file does something or another for you--honest

instead of

       REM this file does something or another for you--honest

· Send error messages that don't need to be seen by the user to null  devices:  e.g.,

       @DEL temp >nul

· Preserve important output

· Be sure, for example, to let the operator see how a command  like

          
NET  USER   %%a   %%b   /ADD

changed the system.  At the very least, I would let this command's output echo to the console.  Better would be code that generated more descriptive messages, like

    

@NET USER user1 foo /add >nul

          
@IF NOT errorlevel==0 @GOTO addfailed

          
@echo attempt to add user user1 succeeded

          
@GOTO endadd

       
     addfailed:

          
@echo attempt to add user user1 failed


     endadd:

· Manage temporary data carefully

· Avoid the use of temporary files, when possible.

· If there’s a need to create  a temporary file

· Before using the file, use an existence check to avoid overwriting any existing files.  Exit, rather than proceed, if the chosen termporary file exists.

· After using the file, delete it before exiting the script, unless the script is supposed to produce this file as one of its outputs.

